
Data Structure
[bookmark: _Toc422985467]Construction of the Union Histories
The most important part of our data preparation was to harmonize the union histories across the different data sets. A “co-residential union” includes any episodes of living with a partner, irrespective of whether the person is married to that partner or not. In other words: A union is equivalent to living with a partner, regardless of marital status. For direct marriages, the start of the co-residential union is equivalent to the start of marriage. A union is not terminated by marriage. It may continue until the end of that particular co-residential episode. In the Finish register data, the Belgium and French (GGS) and in the German (pairfam) data, it is possible to distinguish the end of living with a partner and the end of marriage to that partner. This is not possible in the data for Spain and Italy. Divorce dates are recorded for marital unions, while the end of living together is not surveyed. For these countries, we assume that the end of a union is equivalent to the end of marriage. A key variable of interest is the order of the “co-residential union” (COH_ORDER). For respondents who remain partnered with the parent of the first born child, the order of the union is “1”. For persons who are single, this variable is set to “0”. For persons who are in a co-residential with a new partner who is (probably) not the parent of the first child, COH_ORDER is set to 2. 
Table A1 provides a data extract from the cleaned co-residential histories for three persons from the data for Belgium. EVENT is the indicator variable for whether the second child is born at the end of an episode. _t0 and _t denote the age of the first child, measured in months. COH is the current partnership status. COH_ORDER is the order of the partnership (either marital or non-marital). UN_ORDER02 is a variable that denotes whether a person is still in the same union as he/she was when the first child was born. As can be depicted from the Table, the first person (ID 17) was with his/her first co-residential partner when the child was born. She separated from that person and had a second child with a new partner. The second person (ID 1234) had the first child with his/her second partner and the second child with a new partner as well. In most cases, the end of one co-residential union and the start of the next union are interrupted by spells of being separated or single. This is, however, not the case for ID 6348. He/she has a second child with a new partner without experiencing any episode of singlehood.

Figure A1: Data extract from co-residential histories
	ID
	_t0
	_t
	COH
	UN_ORDER
	UN_ORDER02
	EVENT

	17
	0
	26
	married
	1
	Union with parent of first child
	0

	17
	26
	44
	separated/widowed
	0
	Single
	0

	17
	44
	66
	separated/widowed
	0
	Single
	0

	17
	66
	82
	married
	2
	New union
	1

	1234
	0
	76
	cohabiting
	2
	Union with parent of first child
	0

	1234
	76
	81
	separated/widowed
	0
	Single
	0

	1234
	81
	112
	cohabiting
	3
	New union
	1

	6348
	0
	56
	married
	1
	Union with parent of first child
	0

	6348
	56
	60
	cohabiting
	2
	New union
	0

	6348
	60
	69
	cohabiting
	2
	New union
	1








[bookmark: _Toc422985468]Variables in Harmonized Data


COH_$		Cohabitation (1=yes, 0=no, -1=missing, -2)[footnoteRef:2] [2:  Missing coding: -1: Non-response; System Missing (“.”): Not applicable.] 

COH_Y$		Year start of cohabitation
ICOH_M$	Month start of cohabitation (imputed randomly, if missing)[footnoteRef:3] [3:  Missing month should be imputed by a random number.  

] 


SEP_$		Separation (0=ongoing, 1=separation, 2=death, -1=missing)
SEP_Y$		Year end of cohabitation
ISEP_M$		Month end of cohabitation (imputed randomly, if missing)

MAR_$		Marriage (1=yes, 0=no, -1=missing)
MAR_Y$		Year of marriage
IMAR_M$	Month of marriage (imputed randomly, if missing)

DIV_$		Divorce (0=ongoing, 1=separation, 2=death, -1=missing)
DIV _Y$		Year of divorce
IDIV_M$		Month of divorce (imputed randomly, if missing)
 
KID_$		Birth of biological child (1=yes, 0=no, -1=missing)
KID _Y$		Year of birth
IKID_M$		Month of birth (imputed randomly, if missing)

BORN_Y		Year of Birth
BORN_M		Month of Birth
YEAR_S		Year of Survey

IMONTH_S 	Month of Interview (imputed if missing)

SEX		Sex of Respondent (1: Male; 2: Female)

EDU		Education by ISCED1997 (grouped into 3 categories: 0-2/3-4/5-6)

MIG		Migration Background (	0: Born in country; 1: Not born in country; -1: Missing)

AWGT		Probability weight






[bookmark: _Toc422985469]Stata Code

clear 
set seed 10000
set more off
clear
global WRITE YOURPATH
use DATA.dta, clear
			

*Converts all variables into upper cases (if lower cases due to convserison from SAS)
rename _all, upper

*DELETE CASES: Children born in 1990 or later
keep if KID_Y1>=1990 & KID_Y1<2015

*DELETE CASES: Invalid biographical information
	drop if KID_Y1==-1
	forvalues i=1/20 {
	cap drop if COH_`i'==-1
	cap drop if MAR_`i'==-1
	cap drop if SEP_`i'==-1
	cap drop if DIV_`i'==-1
	cap drop if COH_Y`i'==-1
	cap drop if MAR_Y`i'==-1
	cap drop if SEP_Y`i'==-1
	cap drop if DIV_Y`i'==-1
	cap drop if COH_Y`i'>SEP_Y`i' & COH_Y`i'!=. & SEP_Y`i'!=.
	cap drop if MAR_Y`i'>DIV_Y`i' & MAR_Y`i'!=. & DIV_Y`i'!=.
	}
	
*DELETE CASES: Drop Twins
	drop if ym(KID_Y2,IKID_M2)==ym(KID_Y1,IKID_M1) & KID_2==1
	
*DELETE CASES: Second Child at date of censoring
	drop if ym(YEAR_S,IMONTH_S)==ym(KID_Y2,IKID_M2) & KID_2==1

*DELETE CASES: Drop if R's year of birth is missing
	drop if BORN_Y==-1	
	
*STSETTING THE DATA
	g TIME=ym(YEAR_S,IMONTH_S)-ym(BORN_Y,IBORN_M)
	g TEST=0
	stset TIME, fail(TEST==0) id(ID)


********************************************************************************
********************************************************************************
********************************************************************************
**FERTILITY HISTORIES
	g DUR_KID1=9999
	g DUR_KID2=9999
	replace DUR_KID1=ym(KID_Y1,IKID_M1)-ym(BORN_Y,IBORN_M) if KID_1==1
	replace DUR_KID2=ym(KID_Y2,IKID_M2)-ym(BORN_Y,IBORN_M) if KID_2==1
	stsplit KID1, after at(0) (time=DUR_KID1) 
	stsplit KID2, after at(0) (time=DUR_KID2) 
	recode KID1 -1=0 0=1
	recode KID2 -1=0 0=1
	lab var KID1 "First Child 0=no 1=yes"
	lab var KID2 "Second Child 0=no 1=yes"

	
********************************************************************************
********************************************************************************
********************************************************************************
***PARTNERSHIP HISTORIES
   	forvalues i=1/20 {	

	*Cohabitation
	cap g DUR_COH`i'=9999
	cap replace DUR_COH`i'=ym(COH_Y`i',ICOH_M`i')-ym(BORN_Y,IBORN_M) if COH_`i'==1 
	stsplit COH`i', after at(0) (time=DUR_COH`i') 
	
	*Marriage
	cap g DUR_MAR`i'=9999
	cap replace DUR_MAR`i'=ym(MAR_Y`i',IMAR_M`i')-ym(BORN_Y,IBORN_M) if MAR_`i'==1 
	stsplit MAR`i', after at(0) (time=DUR_MAR`i') 					
	
	*Separation
	cap g DUR_SEP`i'=9999
	cap replace DUR_SEP`i'=ym(SEP_Y`i',ISEP_M`i')-ym(BORN_Y,IBORN_M) if (SEP_`i'==1 | SEP_`i'==2) & SEP_Y`i'!=.a
	cap replace DUR_SEP`i'=45*12 if (SEP_`i'==1 & SEP_Y`i'==.a)
	stsplit SEP`i', after at(0) (time=DUR_SEP`i') 
	
	*Divorce
	cap g DUR_DIV`i'=9999
	cap replace DUR_DIV`i'=ym(DIV_Y`i',IDIV_M`i')-ym(BORN_Y,IBORN_M) if (DIV_`i'==1  & DIV_Y`i'!=.a)
	cap replace DUR_DIV`i'=ym(SEP_Y`i',ISEP_M`i')-ym(BORN_Y,IBORN_M) if (DIV_`i'==1 & DIV_Y`i'==.a & SEP_Y`i'!=.a)
	cap replace DUR_DIV`i'=45*12 if (DIV_`i'==1  & DIV_Y`i'==.a & SEP_Y`i'==.a)
	stsplit DIV`i', after at(0) (time=DUR_DIV`i') 
	}
	

*ORDER OF UNION
	g COH=0
	forvalues i=1/20 {		
	cap replace COH=1 if COH`i'==0 
	cap replace COH=3 if SEP`i'==0 | DIV`i'==0
	cap replace COH=2 if MAR`i'==0
	}
	lab def COH 0 "single" 1 "cohabiting" 2 "married" 3 "separated/widowed" 
	lab val COH COH

	
	g UN_ORDER=0
	forvalues i=1/20 {		
	cap replace UN_ORDER=`i' if COH`i'==0 & (COH==1 | COH==2)
	}
	
*ADJUST PARTNERSHIP ORDER AND COHABITATION STATUS FOR THOSE WHO MOVE TOGETHER WITH PARTNER WITHIN 6 MONTHS AFTER CHILDBIRTH	
	stsplit CORRECT, after at(0,1,2,3,4,5,6) (time=DUR_KID1) 
	sort ID _t0
	by ID: replace UN_ORDER=UN_ORDER[_n+6] if UN_ORDER==0 & CORRECT==0 & UN_ORDER[_n+6]!=0
	by ID: replace UN_ORDER=UN_ORDER[_n+5] if UN_ORDER==0 & CORRECT==1 & UN_ORDER[_n+5]!=0
	by ID: replace UN_ORDER=UN_ORDER[_n+4] if UN_ORDER==0 & CORRECT==2 & UN_ORDER[_n+4]!=0
    by ID: replace UN_ORDER=UN_ORDER[_n+3] if UN_ORDER==0 & CORRECT==3 & UN_ORDER[_n+3]!=0
	by ID: replace UN_ORDER=UN_ORDER[_n+2] if UN_ORDER==0 & CORRECT==4 & UN_ORDER[_n+2]!=0
	by ID: replace UN_ORDER=UN_ORDER[_n+1] if UN_ORDER==0 & CORRECT==5 & UN_ORDER[_n+1]!=0
	
	by ID: replace COH=COH[_n+6] if UN_ORDER==0 & CORRECT==0 & UN_ORDER[_n+6]!=0
	by ID: replace COH=COH[_n+5] if UN_ORDER==0 & CORRECT==1 & UN_ORDER[_n+5]!=0
	by ID: replace COH=COH[_n+4] if UN_ORDER==0 & CORRECT==2 & UN_ORDER[_n+4]!=0
    by ID: replace COH=COH[_n+3] if UN_ORDER==0 & CORRECT==3 & UN_ORDER[_n+3]!=0
	by ID: replace COH=COH[_n+2] if UN_ORDER==0 & CORRECT==4 & UN_ORDER[_n+2]!=0
	by ID: replace COH=COH[_n+1] if UN_ORDER==0 & CORRECT==5 & UN_ORDER[_n+1]!=0

	
	
*TYPE OF CHILDREN
	sort ID _t0 
	g TEST01=-1
	by ID: replace TEST01=UN_ORDER if KID1==0 & KID1[_n+1]==1
	by ID: egen UNORDER_KID01=max(TEST01)
	lab var UNORDER_KID01 "Union order at birth of first child (fixed)"
	
	g UN_ORDER02=.
	replace UN_ORDER02=1 if UNORDER_KID01==UN_ORDER
	replace UN_ORDER02=2 if UNORDER_KID01!=UN_ORDER
	replace UN_ORDER02=0 if COH==0 | COH==3
	lab var UN_ORDER02 "Current union order (time-varying)"
	lab def UN_ORDER02 0 "single" 1 "with parent of first child" 2 "new union" 
	lab val UN_ORDER02 UN_ORDER02
	save $WRITE\FINAL,replace

********************************************************************************
********************************************************************************
********************************************************************************
*Table: Mean Difference between First and Second Birth
	use $WRITE\FINAL,clear
	drop if KID1==0
	by ID: g EVENT=1 if KID2==0 & KID2[_n+1]==1
	keep if EVENT==1 
	generate DUR=(ym(KID_Y2,IKID_M2)-ym(KID_Y1,IKID_M1))/12
	mean DUR if UN_ORDER02==1 [pweight=AWGT] 
	mean DUR if UN_ORDER02!=1 [pweight=AWGT] 
	mean DUR if UN_ORDER02==2 [pweight=AWGT] 
	mean DUR if UN_ORDER02==0 [pweight=AWGT] 
	
********************************************************************************
********************************************************************************
********************************************************************************
*Table: Separation Probabilities by Age of First Child	
	use $WRITE\FINAL,clear
	drop if KID1==0
	sort ID _t0
	
	*Generate Event-variable
	g EVENT=0
	by ID: replace EVENT=1 if (COH==3 | COH==0) & COH[_n-1]==2
	by ID: replace EVENT=1 if (COH==3 | COH==0) & COH[_n-1]==1
	by ID: replace EVENT=1 if (COH==3 | COH==0) & COH[_n-1]==. 
	
	*Generate Time-variable
	generate TEST05=_t if EVENT==1
	by ID: egen TIME_eve=min(TEST05)
	by ID: egen TIME_max=max(_t)
	by ID: egen TIME_min=min(_t0)
	by ID: egen EVENT_max=max(EVENT)
	
	generate TIME_SEP=.
	replace TIME_SEP=TIME_eve-TIME_min 	if EVENT_max==1
	replace TIME_SEP=TIME_max-TIME_min 	if EVENT_max==0

	bysort ID: g SPELL01=_n
	keep if SPELL01==1
	replace TIME_SEP=0.002 if SPELL==1 & EVENT==1
	
	stset TIME_SEP [pweight=AWGT], fail(EVENT_max)
    sts list, at(0.003,12,24,36,48,60,72,84,96,108,120,132,144)

	
********************************************************************************
********************************************************************************
********************************************************************************
*Table: Separation and Second Births as Competing Event	
	use $WRITE\FINAL,clear
	drop if KID1==0
	sort ID _t0

	*Event01: Separation
	g EVENT01=0
	by ID: replace EVENT01=1 if (COH==3 | COH==0) & COH[_n-1]==2
	by ID: replace EVENT01=1 if (COH==3 | COH==0) & COH[_n-1]==1
	by ID: replace EVENT01=1 if (COH==3 | COH==0) & COH[_n-1]==. 
	
	*Event02: Second Birth
	g EVENT02=0
	by ID: replace EVENT02=1 if KID2==0 & KID2[_n+1]==1
	
	
	*Generate Time-variable
	generate TEST05=_t0 if EVENT01==1
	generate TEST06=_t if EVENT02==1
	by ID: egen TIME_event1=min(TEST05)
	by ID: egen TIME_event2=min(TEST06)
	by ID: egen TIME_min=min(_t0)
	by ID: egen DUR=max(_t)
	replace DUR=TIME_event1 if TIME_event1<TIME_event2 & TIME_event2!=. & TIME_event1!=. 
	replace DUR=TIME_event2 if TIME_event2<=TIME_event1 & TIME_event2!=. & TIME_event1!=. 
	replace DUR=TIME_event2  if TIME_event2!=. & TIME_event1==. 
	replace DUR=TIME_event1  if TIME_event1!=. & TIME_event2==. 
	replace DUR=DUR-TIME_min
	
	generate EVENT=0
	replace EVENT=1 if TIME_event1<TIME_event2  & TIME_event2!=. & TIME_event1!=. 
	replace EVENT=2 if TIME_event2<=TIME_event1 & TIME_event2!=. & TIME_event1!=. 
	replace EVENT=2 if TIME_event2!=. & TIME_event1==. 
	replace EVENT=1 if TIME_event1!=. & TIME_event2==. 
	
	bysort ID: g SPELL=_n
	keep if SPELL==1
	replace DUR=0.002 if DUR==0 & EVENT>0
	
	
	*Cumulative Incidence Curves 
	stset DUR [pweight=AWGT], fail(EVENT==1) id(ID)
	tab EVENT
	cap drop Cum* 
	stcompet SURVIVAL=ci , compet1(2)
	sort _t SURVIVAL
	by _t SURVIVAL: g SPELL01=_n
	keep if SPELL01==1
	keep SURVIVAL   _t EVENT
	replace _t=0 if _t<1
	save $WRITE\JUNK, replace	
	
	
	*Cosmetics: Make Output Look Nice
	clear
	set obs 120
	g _t=_n
	save $WRITE\BLANKFILE, replace	

	use $WRITE\JUNK, clear
	keep if EVENT==1
	rename SURVIVAL SURVIVAL01
	save $WRITE\JUNK02, replace	
	use $WRITE\JUNK, clear
	keep if EVENT==2
	merge _t using $WRITE\JUNK02, sort
	drop _merge EVENT
	merge _t using $WRITE\BLANKFILE	, sort
	sort _t
	replace SURVIVAL=SURVIVAL[_n-1] if SURVIVAL[_n-1]!=. &  SURVIVAL==.
	replace SURVIVAL01=SURVIVAL01[_n-1] if SURVIVAL01[_n-1]!=. &  SURVIVAL01==.
	recode SURVIVAL .=0
	recode SURVIVAL01 .=0
	l _t SURVIVAL* if  _t==0 | _t==60 | _t==120 , clean noobs
	



	
********************************************************************************
********************************************************************************
********************************************************************************
*Figure: Transition Probabilities To Second Child
	use $WRITE\FINAL,clear
	drop if KID1==0
	sort ID _t0
	*Generate Event-variable
	sort ID  _t0
	g EVENT=0
	by ID: replace EVENT=1 if KID2==0 & KID2[_n+1]==1
	replace 	   EVENT=2 if EVENT==1 & UN_ORDER02!=1
	
	
	*Keep all at risk of second birth & adjust time variable
	by ID: drop if KID2==1 & EVENT==0 
	by ID: egen TEST03=min(_t0)
	g START=_t0-TEST03
	g END=_t-TEST03
	
	*Cumulative Incidence Curves I
	stset END [pweight=AWGT], fail(EVENT==1) time0(START) id(ID)
	cap drop Cum* 
	stcompet SURVIVAL=ci , compet1(2)
	sort _t SURVIVAL
	by _t SURVIVAL: g SPELL01=_n
	keep if SPELL01==1
	keep SURVIVAL   _t EVENT
	replace _t=0 if _t<1
	save $WRITE\JUNK, replace	

	*Cosmetics: Make Output Look Nice
	clear
	set obs 120
	g _t=_n
	save $WRITE\BLANKFILE, replace	
	use $WRITE\JUNK, clear
	keep if EVENT==1
	rename SURVIVAL SURVIVAL01
	save $WRITE\JUNK02, replace	
	use $WRITE\JUNK, clear
	keep if EVENT==2
	merge _t using $WRITE\JUNK02, sort
	drop _merge EVENT
	merge _t using $WRITE\BLANKFILE	, sort
	sort _t
	replace SURVIVAL=SURVIVAL[_n-1] if SURVIVAL[_n-1]!=. &  SURVIVAL==.
	replace SURVIVAL01=SURVIVAL01[_n-1] if SURVIVAL01[_n-1]!=. &  SURVIVAL01==.
	recode SURVIVAL .=0
	recode SURVIVAL01 .=0
	l _t SURVIVAL01 SURVIVAL, clean noobs
	line SURVIVAL* _t

	
********************************************************************************
********************************************************************************
********************************************************************************	
*Table: Multivariate Analysis	
	use $WRITE\FINAL,clear
	drop if KID1==0
	drop if EDU==-1 | EDU==.
	sort ID _t0
	*Generate Event-variable
	g EVENT=0
	by ID: replace EVENT=1 if KID2==0 & KID2[_n+1]==1

	*Keep all at risk of second birth & adjust time variable
	by ID: drop if KID2==1 & EVENT==0 
	by ID: egen TEST03=min(_t0)
	g START=_t0-TEST03
	g END=_t-TEST03
	stset END, fail(EVENT) time0(START) id(ID)

	*VAR: Age at First Birth
	generate AGE_FIRST=(ym(KID_Y1,IKID_M1)-ym(BORN_Y,IBORN_M))/12
	generate AGE_KID1=1
	replace AGE_KID1=2 if AGE_FIRST>=22
	replace AGE_KID1=3 if AGE_FIRST>=24
	replace AGE_KID1=4 if AGE_FIRST>=26
	replace AGE_KID1=5 if AGE_FIRST>=30
	replace AGE_KID1=6 if AGE_FIRST>=35
	
	*VAR: Age of First Child
	stsplit DUR, at(24, 48, 72, 96, 120)
	
	*Model
	streg ib24.DUR ib1.AGE_KID1 ib1.EDU ib1.UN_ORDER02 ib1.SEX, d(e)
	streg ib24.DUR ib1.AGE_KID1 ib1.EDU#ib2.UN_ORDER02 ib1.SEX, d(e)

	
*Table: Occurrence and Exposure Table	
	stptime,by(DUR) 
	stptime,by(AGE_KID1)
	stptime,by(EDU)
	stptime,by(UN_ORDER02)
	stptime,by(SEX)
	log close

